
model{
  # This is a JAGS script for the following manuscript
  # Title: "Stream resource gradients drive consumption rates of supplemental prey in the ajacent riparian 
zone"
  # Journal: Ecosystems
  # Author: Terui A, Negishi JN, Watanabe N, Nakamura F
  
  ninfo <- 1.0E-4
  lim <- 100
  
  ###Prior###--------------------------------------------------------------------------
  ##regression parameters
  for(i in 1:4){ tau[i] <- pow(sigma[i],-2); sigma[i] ~ dnorm(0,ninfo)T(0,lim) }
  for(i in 1:2){ b.a[i] ~ dnorm(0,ninfo) }
  for(i in 1:2){ b.e[i] ~ dnorm(0,ninfo) }
  for(i in 1:5){ b.gb[i] ~ dnorm(0,ninfo) }
  for(i in 1:3){ b.m[i] ~ dnorm(0,ninfo) }
  
  ##random river-identity effect
  for(m in 1:Nr){ RE[m,1:3] ~ dmnorm(Zero[],TAU[,]) }
  TAU[1:3,1:3] ~ dwish(W[,],4)
  SIGMA[1:3,1:3] <- inverse(TAU[,])
  for(k in 1:3){ st.sigma[k] <- sqrt(SIGMA[k,k]) }
  
  for(i in 1:2){ tau.R[i] <- pow(sigma.R[i],-2); sigma.R[i] ~ dnorm(0,ninfo)T(0,lim) }
  
  ###Likelihood###-----------------------------------------------------------------------------
  ##mean AFDM across seasons----
  for(j in 1:Nx){
    log.afdm_f[j] ~ dnorm(mu.a[j],tau[1])
    mu.a[j] <- b.a[1] + b.a[2]*(log.afdm_s[j]-mean(log.afdm_s[]))
    
    log.x[j] <- (log.afdm_f[j] + log.afdm_s[j])*0.5 #geometric mean
  }
  
  ##primary to secondary production----
  var.e <- pow(sd(fit.e[]), 2)
  R2c.e <- (var.e + SIGMA[1,1])/(var.e + SIGMA[1,1] + pow(sigma[2],2))
  
  for(j in 1:Nx){
    log.e[j] ~ dnorm(log.mu.e[j],tau[2]) # log.e = log.ef in the main text
    log.mu.e[j] <- b.e[1] + b.e[2]*(log.x[j]-mean(log.x[])) + RE[riverID[j],1]
    
    fit.e[j] <- b.e[2]*(log.x[j]-mean(log.x[]))
  }
  
  ##secondary to beetles----
  dsv.gb <- log((1/exp(b.gb[1]))+1)
  var.gb <- pow(sd(fit.gb[]), 2)
  denom <- (var.gb + pow(sd(st.gb[,]),2) + pow(sigma[3],2) + dsv.gb)
  R2c.gb <- (var.gb + pow(sd(st.gb[,]),2))/denom
  
  for(j in 1:Nx){
    for(k in 1:2){
      st.gb[j,k] ~ dnorm(st.mu.gb[j,k], tau.R[1])
      st.mu.gb[j,k] <- b.gb[1] + b.gb[2]*(log.mu.e[j]-mean(log.mu.e[])) + b.gb[3]*(log.pan[j]-
mean(log.pan[])) + b.gb[4]*(area[j,k]-mean(area[,])) + RE[riverID[j],2]
    }
    s.gb[j] <- mean(st.gb[j,])
  }
  
  for(i in 1:Nb){
    beetle[i] ~ dpois(obs.gb[i]) # beetle = GB in the main text
    log(obs.gb[i]) <- log.obs.gb[i]
    log.obs.gb[i] ~ dnorm(log.mu.gb[i], tau[3])
    log.mu.gb[i] <- st.gb[SIDgb[i],year_gb[i]] + b.gb[5]*(temp_gb[i]-mean(temp_gb[])) + log(cpue[i])
    
    fit.gb[i] <- b.gb[5]*(temp_gb[i]-mean(temp_gb[]))
  }
  
  ##beetles to consumption----
  dsv.m <- log((1/exp(b.m[1]))+1)
  var.m <- pow(sd(fit.m[]),2)
  R2c.m <- (var.m + pow(sd(st.m[,]),2))/(var.m + pow(sd(st.m[,]),2) + pow(sigma[4],2) + dsv.m)
  
  for(j in 1:Nx){
    for(k in 1:2){
      st.m[j,k] ~ dnorm(st.mu.m[j,k],tau.R[2])
      st.mu.m[j,k]  <- b.m[1] + b.m[2]*(st.gb[j,k]-mean(st.gb[,])) + RE[riverID[j],3]
    }
  }



  
  
  for(i in 1:Nm){
    isCensored[i] ~ dinterval(Meal[i],mLimit[i]) # Meal = CR in the main text
    Meal[i] ~ dpois(obs.m[i])
    log(obs.m[i]) <- log.obs.m[i]
    log.obs.m[i] ~ dnorm(log.mu.m[i],tau[4])
    log.mu.m[i] <- st.m[SIDm[i],year_m[i]] + b.m[3]*(temp_m[i]-mean(temp_m[])) + log(mtime[i])
    
    fit.m[i] <- b.m[3]*(temp_m[i]-mean(temp_m[]))
  }
  
  ###Parameter scaling###------------------------------------------------------------
  ##primary to secondary production
  sd.e <- sd(log.x[])
  scl_b.e[1] <- b.e[1]; scl_b.e[2] <- b.e[2]*sd.e
  
  ##secondary to beetles
  sd.gb[1] <- sd(log.mu.e[]); sd.gb[2] <- sd(log.pan[])
  sd.gb[3] <- sd(temp_gb[]); sd.gb[4] <- sd(area[,])
  scl_b.gb[1] <- b.gb[1]
  for(i in 1:4){ scl_b.gb[i+1] <- b.gb[i+1]*sd.gb[i] }
  
  ##beetles to consumption
  sd.m[1] <- sd(st.gb[,]); sd.m[2] <- sd(temp_m[])
  scl_b.m[1] <- b.m[1]
  for(i in 1:2){ scl_b.m[i+1] <- b.m[i+1]*sd.m[i] }
}


